Introduction to the Altera Nios Il Soft Processor

This tutorial presents an introduction to Altera’s NBsiI processor, which is a soft processor that can be in-
stantiated on an Altera FPGA device. It describes the baskdtacture of Nios Il and its instruction set. The Nios
Il processor and its associated memory and peripheral coemte are easily instantiated by using Altera’s SOPC
Builder in conjuction with the Quart@ Il software.

A full desciption of the Nios Il processor is provided in thios Il Processor Reference Handboetkhich
is available in the literature section of the Altera web.sin introduction to the SOPC Builder is given in the
tutorial Introduction to the Altera SOPC Buildewhich can be found in the University Program section of tiedw
site.

Altera’s Nios Il is a soft processor, defined in a hardwarecdpson language, which can be implemented in
Altera’s FPGA devices by using the Qua@sl CAD system. This tutorial provides a basic introductiorthe
Nios Il processor, intended for a user who wishes to impleraétios 1l based system on the Altera DE2 board.

1 Nios Il System

The Nios Il processor can be used with a variety of other carapts to form a complete system. These compo-
nents include a number of standard peripherals, but it ®@dssible to define custom peripherals. Altera’s DE2
Development and Education board contains several compotigat can be integrated into a Nios Il system. An
example of such a system is shown in Figure 1.

Figure 1. A Nios Il system implemented on the DE2 board.



The Nios Il processor and the interfaces needed to connetti¢o chips on the DE2 board are implemented in
the Cyclone Il FPGA chip. These components are intercorddzy means of the interconnection network called
the Avalon Switch Fabric. Memory blocks in the Cyclone Il @&vcan be used to provide an on-chip memory for
the Nios Il processor. They can be connected to the procegber directly or through the Avalon network. The
SRAM and SDRAM memory chips on the DE2 board are accessedghrbe appropriate interfaces. Input/output
interfaces are instantiated to provide connection to tBediévices used in the system. A special JTAG UART
interface is used to connect to the circuitry that providemaersal Serial Bus (USB) link to the host computer to
which the DE2 board is connected. This circuitry and the @iased software is called tHéSB-Blaster Another
module, called the JTAG Debug module, is provided to alloevhibst computer to control the Nios Il processor.
It makes it possible to perform operations such as downtmpgrograms into memory, starting and stopping
execution, setting program breakpoints, and collectiadrtiene execution trace data.

Since all parts of the Nios Il system implemented on the FP&ip are defined by using a hardware description
language, a knowledgeable user could write such code tceimgaht any part of the system. This would be an
onnerous and time consuming task. Instead, one can use A€ B0ilder tool in the Quartus Il software to
implement a desired system simply by choosing the requioetponents and specifying the parameters needed
to make each component fit the overall requirements of thesys

2 Overview of Nios Il Processor Features

The Nios Il processor has a number of features that can begcoeéi by the user to meet the demands of a desired
system. The processor can be implemented in three diffecarfitgurations:

e Nios lI/f is a "fast" version designed for superior perfoma. It has the widest scope of configuration
options that can be used to optimize the processor for padoce.

e Nios ll/s is a "standard" version that requires less ressiinc an FPGA device as a trade-off for reduced
performance.

¢ Nios ll/e is an "economy" version which requires the leasbant of FPGA resources, but also has the most
limited set of user-configurable features.

The Nios Il processor has a Reduced Instruction Set ComRt&C) architecture. Its arithmetic and logic
operations are performed on operands in the general purpgsters. The data is moved between the memory
and these registers by means of Load and Store instructions.

The wordlength of the Nios Il processor is 32 bits. All registare 32 bits long. Byte addresses in a 32-bit
word are assigned ilittle-endianstyle, in which the lower byte addresses are used for theslgsas#ficant bytes
(the rightmost bytes) of the word.

The Nios Il architecture uses separate instruction and loiz&as, which is often referred to as tHarvard
architecture.

A Nios Il processor may operate in the following three modes:

e Supervisor mode allows the processor to execute all instructions and perédl available functions. When
the processor is reset, it enters this mode.

e User mode- the intent of this mode is to prevent execution of some lesions that shoud be used for
systems purposes only. Some processor features are nesdnbeén this mode.

e Debug mode- is used by software debugging tools to implement featureb as breakpoints and watch-
points.

Application programs can be run in either the User or Sugervinodes. Presently available versions of the Nios
Il processor do not support the User mode.



3 Register Structure

The Nios Il processor has thirty two 32-bit general purpeggsters, as shown in Figure 2. Some of these registers
are intended for a specific purpose and have special nantesr¢h@cognized by the Assembler.

e RegistenO is referred to as theeroregister. It always contains the constant 0. Thus, readtisgrégister
returns the value 0, while writing to it has no effect.

o Registerl is used by the Assembler as a temporary register; it shodldeneferenced in user programs
e Registerg24 andr29 are used for processing of exceptions; they are not availalllser mode

o Registerg25 andr30 are used exclusively by the JTAG Debug module

e Registerg27 andr28 are used to control the stack used by the Nios Il processor

¢ Register31is used to hold the return address when a subroutine is called

Register | Name | Function

ro zero | 0x00000000

rl at Assembler Temporary

r2

r3

r23

r24 et Exception Temporaryl)

r25 bt Breakpoint Temporar{?)

r26 ap Global Pointer

r27 sp Stack Pointer

r28 fp Frame Pointer

r29 ea Exception Return Addreg4)
r30 ba Breakpoint Return Addreg®)
r31 ra Return Address

(1) The register is not available in User mode
(2) The register is used exclusively by the JTAG Debug module

Figure 2. General Purpose registers.

There are six 32-bit control registers, as indicated in FigRi The names given in the figure are recognized
by the Assembler. These registers are used automaticaltyfdrol purposes. They can be read and written to by
special instructionsdctl andwrctl, which can be executed only in the supervisor mode. Thetszgiare used as
follows:

o RegisterctlO reflects the operating status of the processor. Only twaobitisis register are meaningful:

— U is the User/Supervisor mode bif; = 1 for User mode, whilé/ = 0 for Supervisor mode.
— PIE is the processor interrupt-enable bit. WHeIE = 1, the processor may accept external interrupts.
WhenPIE = 0, the processor ignores external interrupts.

¢ Registerctll holds a saved copy of the status register during exceptiocessing. The bitEU andEPIE
are the saved values of the status bitandPIE.



¢ Registerctl2 holds a saved copy of the status register during debug breakgsing. The bitBU andBPIE
are the saved values of the status bitandPIE.

¢ Registerctl3 is used to enable individual external interrupts. Each diitesponds to one of the interrupts
irg0 to irg31. The value of 1 means that the interrupt is enabled, while @a¢hat it is disabled.

¢ Registerctl4 indicates which interrupts are pending. The value of a ghienti4y, is setto 1 if the interrupt
irgk is both active and enabled by having the interrupt-enate®8;, set to 1.

¢ Registerctl5 holds a value that uniquely identifies the processor in aiproltessor system.

Register | Name bs1 o+ bo | by bo
ctlo status Reserved U PIE
ctll estatus Reserved | EU | EPIE
ctl2 bstatus Reserved | BU | BPIE
ctl3 ienable Interrupt-enable bits
ctl4 ipending Pending-interrupt bits
ctl5 cpuid Unique processor identifier

Figure 3. Control registers.

4 Accessing Memory and 1/0O Devices

Figure 4 shows how a Nios Il processor can access memory @mikelVices. For best performance, the Nios Il/f
processor can include both instruction and data cachescadhes are implemented in the FPGA memory blocks.
Their usage is optional and they are specified (includingr #iee) at the system generation time by using the
SOPC Builder. The Nios Il/s version can have the instructiache but not the data cache. The Nios ll/e version
has neither instruction nor data cache.

Another way to give the processor fast access to the on-chipary is by using théghtly coupledmemory
arrangement, in which case the processor accesses the yneimar direct path rather than through the Avalon
network. Accesses to a tightly coupled memory bypass theecaemory. There can be one or more tightly
coupled instruction and data memories. If the instructiache is not included in a system, then there must be at
least one tightly coupled memory provided for Nios Il/f antb#ll/s processors. On-chip memory can also be
accessed via the Avalon network.

Off-chip memory devices, such as SRAM, SDRAM, and Flash nrgrobips are accessed by instantiating the
appropriate interfaces. The input/output devices are mgmapped and can be accessed as memory locations.

Data accesses to memory locations and I/O interfaces dierped by means of Load and Store instructions,
which cause data to be transferred between the memory aeda@urpose registers.

Figure 4. Memory and I/O organization.

5 Addressing

The Nios Il processor issues 32-bit addresses. The memaogesp byte-addressable. Instructions can read and
write words(32 bits),halfwords(16 bits), orbytes(8 bits) of data. Reading or writing to an address that doés no
correspond to an existing memory or 1/O location producesratefined result.

There are five addressing modes provided:

¢ Immediate mode a 16-bit operand is given explicitly in the instruction.iFkalue may be sign extended
to produce a 32-bit operand in instructions that perforharetic operations.

e Register mode the operand is in a processor register

e Displacement mode the effective address of the operand is the sum of the cantéra register and a
signed 16-bit displacement value given in the instruction

e Register indirect mode the effective address of the operand is the contents ofisteegpecified in the
instruction. This is equivalent to the displacement modenethe displacement value is equal to 0.



e Absolute mode a 16-bit absolute address of an operand can be specifiedrgythe displacement mode
with registerrO which always contains the value O.

6 Instructions

All Nios Il instructions are 32-bits long. In addition to nfdoe instructions that are executed directly by the pro-
cessor, the Nios Il instruction set includes a numbgrsgfudoinstructionthat can be used in assembly language
programs. The Assembler replaces each pseudoinstrugtionédbor more machine instructions.

Figure 5 depicts the three possible instruction formatgpg, R-type and J-type. In all cases the six bitsy
denote the OP code. The remaining bits are used to specifteey immediate operands, or extended OP codes.

e |-type — Five-bit fields A and B are used to specify generappae registers. A 16-bit field IMMED16
provides immediate data which can be sign extended to peav8R-bit operand.

e R-type — Five-bit fields A, B and C are used to specify geneuappse registers. An 11-bit field OPX is
used to extend the OP code.

e J-type — A 26-bit field IMMED26 contains an unsigned immeeligalue. This format is used only in the
Call instruction.

31 27 26 22 21 6 5 0
A B IMMED16 OoP
(a) I-type
31 27 26 22 21 17 16 6 5 0
A B C OPX OoP
(b) R-type
31 6 5 0
IMMED26 OoP
(c) J-type

Figure 5. Formats of Nios Il instructions.

The following subsections discuss briefly the main featufethe Nios Il instruction set. For a complete
description of the instruction set, including the detaifshow each instruction is encoded, the reader should
consult theNios Il Processor Reference Handbook

6.1 Load and Store Instructions

Load and Store instructions are used to move data betweeroméend 1/0 interfaces) and the general purpose
registers. They are of I-type. For example, the Load Worttiiction

ldw rB, byte_offset(rA)



determines the effective address of a memory location asutimeof a byte offset value and the contents of register
A. The 16-bit byte_offset value is sign extended to 32 bite 3&-bit memory operand is loaded into regidier

For instance, assume that the contents of registerre 1260,y and the byte_offset value &),,. Then, the
instruction

ldw r3, 80(r4)

loads the 32-bit operand at memory addres), into registern-3.

The Store Word instruction has the format
stw rB, byte offset(rA)

It stores the contents of registBrinto the memory location at the address computed as the stime biyte_offset
value and the contents of registér

There are Load and Store instructions that use operandarthanly 8 or 16 bits long. They are referred to as
Load/Store Byte and Load/Store Halfword instructionspeesively. Such Load instructions are:

¢ |db (Load Byte)

¢ |dbu (Load Byte Unsigned)

¢ |dh (Load Halfword)

e |dhu (Load Halfword Unsigned)

When a shorter operand is loaded into a 32-bit register, Itsevaas to be adjusted to fit into the register. This
is done by sign extending the 8- or 16-bit value to 32 bits mlto andldh instructions. In thédbu andldhu
instructions the operand is zero extended.

The corresponding Store instructions are:
e stb (Store Byte)
e sth (Store Halfword)

Thestb instruction stores the low byte of registBrinto the memory byte specified by the effective address. The
sth instruction stores the low halfword of regist8r In this case the effective address must be halfword aligned

Each Load and Store instruction has a version intended t@saing locations in I/O device interfaces. These
instructions are:

e |ldwio (Load Word I/O)

e |dbio (Load Byte I/O)

e |dbuio (Load Byte Unsigned 1/O)

e |dhio (Load Halfword I/O)

e |dhuio (Load Halfword Unsigned 1/O)
e stwio (Store Word 1/0O)

e stbio (Store Byte I/0O)

e sthio (Store Halfword I/O)

The difference is that these instructions bypass the c#obre exists.



6.2 Arithmetic Instructions

The arithmetic instructions operate on the data that ieiththe general purpose registers or given as an imme-
diate value in the instruction. These instructions are typ& or I-type, respectively. They include:

e add (Add Registers)
e addi (Add Immediate)
e sub (Subtract Registers)
e subi (Subtract Immediate)
o mul (Multiply)
e muli (Multiply Immediate)
¢ div (Divide)
e divu (Divide Unsigned)
The Add instruction
add rC, rA, rB

adds the contents of registetsand B, and places the sum into registér

The Add Immediate instruction
addi rB, rA, IMMED16

adds the contents of registdrand the sign-extended 16-bit operand given in the instsactind places the result
into registerB. The addition operation in these instructions is the samédth signed and unsigned operands;
there are no condition flags that are set by the operatiors mbians that when unsigned operands are added, the
carry from the most significant bit position has to be dektigexecuting a separate instruction. Similarly, when
signed operands are added, the arithmetic overflow has tetbetdd separately. The detection of these conditions
is dicussed in section 6.11.

The Subtract instruction
sub rC, rA, IB

subtracts the contents of registBrfrom registerA, and places the result into registér Again, the carry and
overflow detection has to be done by using additional intityas, as explained in section 6.11.

The immediate versiorsubi, is a pseudoinstruction implemented as

addi rB, rA, -IMMED16

The Multiply instruction
mul rC, rA, rB

multiplies the contents of registersand B, and places the low-order 32 bits of the product into regiSteThe
operands are treated as unsigned numbers. The carry arfibeveetection has to be done by using additional
instructions. In the immediate version

muli rB, rA, IMMED16

the 16-bit immediate operand is sign extended to 32 bits.



The Divide instruction
div rC, rA, 1B

divides the contents of register by the contents of registd® and places the integer portion of the quotient into
registerC. The operands are treated as signed integersdiMoeinstruction is performed in the same way except
that the operands are treated as unsigned integers.

6.3 Logic Instructions

The logic instructions provide the AND, OR, XOR, and NOR @tiems. They operate on data that is either in
the general purpose registers or given as an immediate welle instruction. These instructions are of R-type or
I-type, respectively.

The AND instruction
and rC, rA, 1B

performs a bitwise logical AND of the contents of registdrand B, and stores the result in registér Similarly,
the instructionr, xor andnor perform the OR, XOR and NOR operations, respectively.

The AND Immediate instruction
andi rB, rA, IMMED16

performs a bitwise logical AND of the contents of registeand the IMMED16 operand which is zero-extended
to 32 bits, and stores the result in register Similarly, the instructionsri, xori andnori perform the OR, XOR
and NOR operations, respectively.

It is also possible to use the 16-bit immediate operand ag@hegh-order bits in the logic operations, in which
case the low-order 16 bits of the operand are zeros. This@ngglished with the instructions:

e andhi (AND High Immediate)
e orhi (OR High Immediate)
e xorhi (XOR High Immediate)

6.4 Move Instructions

The Move instructions copy the contents of one register amother, or they place an immediate value into a
register. They are pseudoinstructions implemented bygustimer instructions. The instruction

mov rC, rA
copies the contents of registdrinto registerC. It is implemented as

add rC, rA, r0

The Move Immediate instruction
movi B, IMMED16
sign extends the IMMED16 value to 32 bits and loads it intasteg B. It is implemented as
addi rB, r0, IMMED16

The Move Unsigned Immediate instruction

10



movui rB, IMMED16
zero extends the IMMED16 value to 32 bits and loads it intastegB. It is implemented as
ori rB, r0, IMMED16

The Move Immediate Address instruction
movia rB, LABEL
loads a 32-bit value that corresponds to the addt@&EL into registerB. It is implemented as:

orhi  rB, r0, %hi(LABEL)
ori B, rB, %lo(LABEL)

The %hi(LABEL) and%lo(LABEL) are the Assembler macros which extract the high-order Koanitl the low-
order 16 bits, respectively, of a 32-bit valuABEL The orhi instruction sets the high-order bits of register
followed by theori instruction which sets the low-order bits Bf Note that two instructions are used because the
I-type format provides for only a 16-bit immediate operand.

6.5 Comparison Instructions

The Comparison instructions compare the contents of twigtexg or the contents of a register and an immediate
value, and write either 1 (if true) or O (if false) into the uéisegister. They are of R-type or I-type, respectively.
These instructions correspond to the equality and relatioperators in the C programming language.

The Compare Less Than Signed instruction
cmplt rC, rA, rB

performs the comparison of signed numbers in registeasid B, rA < rB, and writes a 1 into register if the
result is true; otherwise, it writes a 0.

The Compare Less Than Unsigned instruction
cmpltu rC, rA, rB

performs the same function as tt@plt instruction, but it treats the operands as unsigned numbers

Other instructions of this type are:
e cmpeq rC, rA, rB (Comparison rA ==rB)
e cmpne rC, rA, rB (Comparison rA !=rB)
e cmpge rC, rA, rB (Signed comparison r& = rB)
e cmpgeu rC, rA, rB (Unsigned comparison r&=rB)

e cmpgt rC, rA, rB (Signed comparison rA- rB)
This is a pseudoinstruction implemented asahwlt instruction by swapping its rA and rB operands.

e cmpgtu rC, rA, rB (Unsigned comparison r& rB)
This is a pseudoinstruction implemented asdhwltu instruction by swapping its rA and rB operands.

e cmple rC, rA, rB (Signed comparison rA=rB)
This is a pseudoinstruction implemented asdhgge instruction by swapping its rA and rB operands.

e cmpleu rC, rA, rB (Unsigned comparison r&= rB)
This is a pseudoinstruction implemented asdhggeu instruction by swapping its rA and rB operands.

The immediate versions of the Comparison instructionsli@e/an immediate operand. For example, the
Compare Less Than Signed Immediate instruction

cmplti rB, rA, IMMED16

compares the signed number in registewith the sign-extended immediate operand. It writes a lrietgsterB
if rA < IMMED16; otherwise, it writes a O.

The Compare Less Than Unsigned Immediate instruction

cmpltui rB, rA, IMMED16

compares the unsigned number in registewith the zero-extended immediate operand. It writes a 1reddister
B if rA < IMMED16; otherwise, it writes a O.

Other instructions of this type are:

6.6

cmpeqi rB, rA, IMMED16 (Comparison rA == IMMED16)

cmpnei rB, rA, IMMED16 (Comparison rA != IMMED16)

cmpgei rB, rA, IMMED16 (Signed comparison r&= IMMED16)
cmpgeui rB, rA, IMMED16 (Unsigned comparison r&=IMMED16)

cmpgti rB, rA, IMMED16 (Signed comparison r& IMMED16)
This is a pseudoinstruction which is implemented by usimgthpgei instruction with an immediate value
IMMED16 + 1.

cmpgtui rB, rA, IMMED16 (Unsigned comparison r& IMMED16)
This is a pseudoinstruction which is implemented by usirggdimpgeui instruction with an immediate
value IMMED16 + 1.

cmplei rB, rA, IMMED16 (Signed comparison rA=IMMED16)
This is a pseudoinstruction which is implemented by usirgrthplti instruction with an immediate value
IMMED16 + 1.

cmpleui rB, rA, IMMED16 (Unsigned comparison r&=IMMED16)
This is a pseudoinstruction which is implemented by usimgthpltui instruction with an immediate value
IMMED16 + 1.

Shift Instructions

The Shift instructions shift the contents of a registeraitto the right or to the left. They are of R-type. They
correspond to the shift operators> and<<, in the C programming language. These instructions are:

srl rC, rA, rB (Shift Right Logical)

srli rC, rA, IMMEDS (Shift Right Logical Immediate)
sra rC, rA, rB (Shift Right Arithmetic)

srai rC, rA, IMMEDS (Shift Right Arithmetic Immediate)
sll rC, rA, rB (Shift Left Logical)

slli rC, rA, IMMEDS5 (Shift Left Logical Immediate)

Thesrl instruction shifts the contents of registérto the right by the number of bit positions specified by the five
least-significant bits (number in the range 0 to 31) in regi8t, and stores the result in register The vacated
bits on the left side of the shifted operand are filled with 0s.

The srli instruction shifts the contents of registdrto the right by the number of bit positions specified by the
five-bit unsigned value, IMMEDS5, given in the instruction.

The sra andsrai instructions perform the same actions as sHeand srli instructions, except that the sign bit,
rAsi, is replicated into the vacated bits on the left side of thteshoperand.

Thesll andslli instructions are similar to therl andsrli instructions, but they shift the operand in registeto the
left and fill the vacated bits on the right side with Os.

6.7 Rotate Instructions

There are three Rotate instructions, which use the R-typeado
e ror rC, rA, rB (Rotate Right)
e rol rC, rA, rB (Rotate Left)
e roli rC, rA, IMMEDS (Rotate Left Immediate)

Theror instruction rotates the bits of registdrin the left-to-right direction by the number of bit posit®epec-
ified by the five least-significant bits (number in the range 8X) in registei3, and stores the result in registér

Therol instruction is similar to theor instruction, but it rotates the operand in the right-td-téfection.

Theroli instruction rotates the bits of registérin the right-to-left direction by the number of bit positespecified
by the five-bit unsigned value, IMMEDS5, given in the instioat and stores the result in registér
6.8 Branch and Jump Instructions

The flow of execution of a program can be changed by executingdh or Jump instructions. It may be changed
either unconditionally or conditionally.

The Jump instruction
jmp rA
transfers execution unconditionally to the address coathin registerA.
The Unconditional Branch instruction
br LABEL

transfers execution unconditionally to the instructiom@tress ABEL This is an instruction of I-type, in which
a 16-bit immediate value (interpreted as a signed numbegifsps the offset to the branch target instruction. The
offset is the distance in bytes from the instruction that idiately followsbr to the addressABEL

Conditional transfer of execution is achieved with the Gtadal Branch instructions, which compare the

contents of two registers and cause a branch if the resuliés These instructions are of I-type and the offset is
determined as explained above for tiranstruction.

The Branch if Less Than Signed instruction
blt rA, rB, LABEL

performs the comparisorA < rB, treating the contents of the registers as signed numbers.

The Branch if Less Than Unsigned instruction
bltu rA, rB, LABEL

performs the comparisorA < rB, treating the contents of the registers as unsigned numbers

The other Conditional Branch instructions are:
e beq rA, rB, LABEL (Comparison rA ==rB)
e bne rA, rB, LABEL (Comparison rA !=rB)
e bge rA, rB, LABEL (Signed comparison rA&=rB)
e bgeu rA, rB, LABEL (Unsigned comparison r&= rB)

e bgt rA, rB, LABEL (Signed comparison rA rB)
This is a pseudoinstruction implemented ashitiénstruction by swapping the register operands.

e bgtu rA, rB, LABEL (Unsigned comparison r£ rB)
This is a pseudoinstruction implemented aslilie instruction by swapping the register operands.

e ble rA, rB, LABEL (Signed comparison rA=rB)
This is a pseudoinstruction implemented ashife instruction by swapping the register operands.

e bleu rA, rB, LABEL (Unsigned comparison rA=rB)
This is a pseudoinstruction implemented ashifeu instruction by swapping the register operands.

6.9 Subroutine Linkage Instructions
Nios Il has two instructions for calling subroutines. ThdlGaibroutine instruction
call LABEL

is of J-type, which includes a 26-bit unsigned immediatai@gMMED26). The instruction saves the return
address (which is the address of the next instruction) irsteg-31. Then, it transfers control to the instruction
at addres& ABEL This address is determined by concatenating the four digbr bits of the Program Counter
with the IMMEDZ26 value as follows

Jump address = PC31_9g : IMMED26 : 00
Note that the two least-significant bits are 0 because Niwstiuctions must be aligned on word boundaries.
The Call Subroutine in Register instruction

callr rA

is of R-type. It saves the return address in registdrand then transfers control to the instruction at the address
contained in registeA.

Return from a subroutine is performed with the instruction
ret

This instruction transfers execution to the address coathin register-31.

6.10 Control Instructions
The Nios Il control registers can be read and written by spé@tstructions. The Read Control Register instruction
rdctl rC, ctiN

copies the contents of control registtiN into registerC'

The Write Control Register instruction
wrctl ctIN, rA

copies the contents of register A into the control registi\.

There are two instructions provided for dealing with exaaps: trap anderet. They are similar to theall
andret instructions, but they are used for exceptions. Their ugésimussed in section 8.2.

The instructionsbreak andbret generate breaks and return from breaks. They are used ietyulsy the
software debugging tools.

The Nios Il cache memories are managed with the instructitushd (Flush Data Cache Lineflushi (Flush
Instruction Cache Line)initd (Initialize Data Cache Line), anihiti (Initialize Instruction Cache Line). These
instructions are discussed in section 9.1.

6.11 Carry and Overflow Detection

As pointed out in section 6.2, the Add and Subtract instomstiperform the corresponding operations in the same
way for both signed and unsigned operands. The possiblg aad arithmetic overflow conditions are not de-
tected, because Nios Il does not contain condition flagsrtfight be set as a result. These conditions can be
detected by using additional instructions.

Consider the Add instruction
add rC, rA, rB

Having executed this instruction, a possible occurrence cérry out of the most-significant bi€’§;) can be
detected by checking whether the unsigned sum (in registas less than one of the unsigned operands. For
example, if this instruction is followed by the instruction

cmpltu rD, rC, rA

then the carry bit will be written into registép.

Similarly, if a branch is required when a carry occurs, tleis be accomplished as follows:

add rC,rA, B
bltu rC, rA, LABEL

A test for arithmetic overflow can be done by checking the sigfithe summands and the resulting sum. An
overflow occurs if two positive numbers produce a negativa,sor if two negative numbers produce a positive
sum. Using this approach, the overflow condition can comtrmnditional branch as follows:

add rC,rA, B /* The required Add operation */

xor 1D, rC, rA [* Compare signs of sum and rA */

xor rE, rC, B [* Compare signs of sum and rB */

and rD, D, rE [*SetD3; = 1if ((Agl == Bgl) = Cgl) */

blt rD, r0, LABEL /* Branch if overflow occurred */

A similar approach can be used to detect the carry and ovechogitions in Subtract operations. A carry out
of the most-significant bit of the resulting difference candetected by checking whether the first operand is less
than the second operand. Thus, the carry can be used tolcetinditional branch as follows:

sub rC, rA, 1B
bltu rA, rB, LABEL

The arithmetic overflow in a Subtract operation is detecteddmparing the sign of the generated difference with
the signs of the operands. Overflow occurs if the operandsgisters4 and B have different signs, and the sign
of the difference in registet' is different than the sign ofl. Thus, a conditional branch based on the arithmetic
overflow can be achieved as follows:

sub rC,rA, 1B [* The required Subtract operation */

xor D, rA, 1B /* Compare signs of rA and rB */

xor rE, rA, rC [* Compare signs of rA and rC */

and rD, D, rE [*SetD3; = 1if ((Agl = Bgl) && (A31 = Cgl)) */

blt rD, r0, LABEL /* Branch if overflow occurred */

7 Assembler Directives

The Nios Il Assembler conforms to the widely used GNU Assa@mbkhich is software available in the public
domain. Thus, the GNU Assembler directives can be used ia Njgrograms. Assembler directives begin with a
period. We describe some of the more frequently used aseentibtctives below.

.ascii "string'"...

A string of ASCII characters is loaded into consecutive taddresses in the memory. Multiple strings, separated
by commas, can be specified.

.asciz "string"...
This directive is the same aascii, except that each string is followed (terminated) by a zete.b
.byte expressions

Expressions separated by commas are specified. Each éaprissassembled into the next byte. Examples of
expressions are: 8, 5 + LABEL, and K 6.

.data

Identifies the data that should be placed in the data sectitreanemory. The desired memory location for the
data section can be specified in the Altera Monitor Prograyssem configuration window.

.end
Marks the end of the source code file; everything after thisatiive is ignored by the assembler.
.equ symbol, expression

Sets the value cdymbolto expression

.global symbol

Makessymbolvisible outside the assembled object file.

.hword expressions

Expressions separated by commas are specified. Each égprissassembled into a 16-bit number.

.include "filenamé&

Provides a mechanism for including supporting files in aserogram.

.org new-Ic

Advances the location counter Imgw-Ig wherenew-Icis used as an offset from the starting location specified

in the Altera Monitor Program’s system configuration winddvwe.org directive may only increase the location
counter, or leave it unchanged; it cannot move the locatiamter backwards.

.skip size
Emits the number of bytes specifiedsize the value of each byte is zero.

text

Identifies the code that should be placed in the text secfidimeomemory. The desired memory location for the
text section can be specified in the Altera Monitor Prograsy&tem configuration window.

.word expressions

Expressions separated by commas are specified. Each égprissassembled into a 32-bit number.

8 Example Program

As an illustration of Nios Il instructions and assembleedtives, Figure 6 gives an assembly language program
that computes a dot product of two vectofsandB. The vectors have elements. The required computation is

Dot product= """ A(i) x B(i)

The vectors are stored in memory locations at addreé®sE€ TORandBVECTORrespectively. The number of el-
ementsn, is stored in memory locatiolV. The computed resultis written into memory locatid®@T_PRODUCT
Each vector element is assumed to be a signed 32-bit number.

.include "nios_macros.s"
.global _start

start:
movia r2, AVECTOR /* Register2 is a pointer to vectod */
movia r3, BVECTOR /* Register3 is a pointer to vectoB */
movia r4, N
ldw  r4, 0(r4) /* Register4 is used as the counter for loop iterations */
add r5,r0,r0 [* Register5 is used to accumulate the product */
LOOP: Idw 6, 0(r2) /* Load the next element of vectdrt/
ldw  r7,0(r3) /* Load the next element of vectdr*/
mul  r8,r6,r7 [* Compute the product of next pair of elemerits *
add r5,r5,r8 /* Add to the sum */
addi r2,r2,4 /* Increment the pointer to vectér/
addi r3,r3,4 /* Increment the pointer to vectBr*/
subi  r4,r4,1 /* Decrement the counter */
bgt r4,r0, LOOP [* Loop again if not finished */

stw 5, DOT_PRODUCT(r0) /* Store the result in memory */
STOP: br STOP

N:

word 6 /* Specify the number of elements */

AVECTOR:

word 5,3,-6,19, 8,12 I* Specify the elements of vector A */
BVECTOR:

word  2,14,-3,2,-5, 36 [* Specify the elements of vector B */
DOT_PRODUCT:

skip 4

Figure 6. A program that computes the dot product of two wscto

Note that the program ends by continuously looping on thieBesnch instruction. If instead we wanted to pass
control to debugging software, we could replace tirinstruction with thebreak instruction.

The program includes the assembler directive
.include "nios_macros.s"

which informs the Assembler to use some macro commands #vatlieen created for the Nios Il processor. In
this program, the macro used converts ti@via pseudoinstruction into two OR instructions as explaineskicr
tion 6.4.

The directive
.global _start

indicates to the Assembler that the labstartis accessible outside the assembled object file. This |aktlei
default label we use to indicate to the Linker program thdregg of the application program.

The program includes some sample data. It illustrates hewnatbrd directive can be used to load data items
into memory. The memory locations involved are those thidviothe location occupied by thier instruction.
Since we have not explicitly specified the starting addréfiseoprogram itself, the assembled code will be loaded
in memory starting at address 0.

To execute the program in Figure 6 on Altera’s DE2 board, itdsessary to implement a Nios Il processor
and its memory (which can be just the on-chip memory of thel@eell FPGA). Since the program includes the

Multiply instruction, it cannot be executed on the econoreysion of the processor, because Nios Il/e does not
support themul instruction. Either Nios II/s or Nios Il/f processors canused.

The tutorialIntroduction to the Altera SOPC Buildexxplains how a Nios Il system can be implemented.
The tutorial Altera Monitor Programexplains how an application program can be assembled, dadetl and
executed on the DE2 board.

9 Exception Processing

An exceptiorin the normal flow of program execution can be caused by:
e Software trap
e Hardware interrupt
e Unimplemented instruction
In response to an exception the Nios Il processor performfaflowing actions:

1. Saves the existing processor status information by ogptyie contents of thstatusregister €tl0) into the
estatugegister €tl1)

. Clears thdJ bit in the statusregister, to ensure that the processor is in the Superviedem
. Clears théIE bit in the statusregister, thus disabling the additional external procesgerrupts

. Writes the address of the instruction after the exceptitmtheearegister (29)

g W0 DN

. Transfers execution to the address ofdkeeption handlewhich determines the cause of the exception and
dispatches an appropriag&ception routingo respond to the exception

The address of the exception handler is specified at systapragon time using the SOPC Builder, and it cannot
be changed by software at run time. This address can be pabigthe designer; otherwise, the default address
is 20,6 from the starting address of the main memory. For exampkieifmemory starts at address 0, then the
default address of the exception handler is 0x00000020.

9.1 Software Trap

A software exception occurs whenti@p instruction is encountered in a program. This instructiaves the
address of the next instruction in tearegister (29). Then, it disables interrupts and transfers executioi¢o t
exception handler.

In the exception-service routine the last instructioerist (Exception Return), which returns execution control
to the instruction that follows thap instruction that caused the exception. The return addsegwén by the
contents of registega Theeret instruction restores the previous status of the processoopying the contents
of theestatugegister into thestatusregister.

A common use of the software trap is to transfer control tdfemdint program, such as an operating system.

9.2 Hardware Interrupts

Hardware interrupts can be raised by external sources,a devices, by asserting one of the processor’'s 32
interrupt-request inputsrg0 throughirg31. An interrupt is generated only if the following three camahis are
true:

e ThePIE bit in thestatusregister is setto 1
e An interrupt-request inputtgk, is asserted

e The corresponding interrupt-enable loitl3;, is setto 1

The contents of th@gendingregister €tl4) indicate which interrupt requests are pending. An exoeptoutine
determines which of the pending interrupts has the highestity, and transfers control to the corresponding
interrupt-service routine

Upon completion of the interrupt-service routine, the esim control is returned to the interrupted program
by means of theeret instruction, as explained above. However, since an extamtexrupt request is handled
without first completing the instruction that is being execlwhen the interrupt occurs, the interrupted instruction
must be re-executed upon return from the interrupt-semdaoéne. To achieve this, the interrupt-service routine
has to adjust the contents of tearegister which are at this time pointing to the next instiarcbf the interrupted
program. Hence, the value in tkaregister has to be decremented by 4 prior to executingitbeinstruction.

9.3 Unimplemented Instructions

This exception occurs when the processor encounters ainatrdction that is not implemented in hardware. This
may be the case with instructions suchmad anddiv. The exception handler may call a routine that emulates the
required operation in software.

9.4 Determining the Type of Exception

When an exception occurs, the exception-handling routisdddetermine what type of exception has occurred.
The order in which the exceptions should be checked is:

1. Read thdpendingregister to see if a hardware interrupt has occurred; if Isen tyo to the appropriate
interrupt-service routine.

2. Read the instruction that was being executed when thepBgneoccurred. The address of this instruction
is the value in theearegister minus 4. If this is thap instruction, then go to the software-trap-handling
routine.

3. Otherwise, the exception is due to an unimplementeducison.

9.5 Exception Processing Example

The following example illustrates the Nios Il code neededdal with a hardware interrupt. We will assume that
an 1/0 device raises an interrupt request on the interrequiest inpuirgl. Also, let the exception handler start at
address 0x020, and the interrupt-service routine foirtifferequest start at address 0x0100.

Figure 7 shows a portion of the code that can be used for tmgoge. The exception handler first determines
the type of exception that has occurred. Having determihatithere is a hardware interrupt request, it finds the
specific interrupt by examining the bits of tleeregister which has a copy of control registai. If bit et; is
equal to 1, then the the interrupt-service routine EXT_IR€®Qéxecuted. Otherwise, it is necessary to check for
other possible interrupts.

Note that in Figure 7 we are using regist&B in the process of testing whether the initl is setto 1. In a
practical application program this register may also baludee some other purpose, in which case its contents
should first be saved on the stack and later restored priettioning from the exception handler.

.org 0x20
[* Exception handler */

rdctl et, ctl4 [* Check if external interrupt occurred */

beq et r0, OTHER_EXCEPTIONS /*If zero, check exceptions */

subi ea, ea, 4 [* Hardware interrupt, decrement ea to exdéleat@terrupted */
/* instruction upon return to main program */

andi rl3,et, 2 /* Check if irql asserted */

bne r13,r0, EXT_IRQ1 /* If yes, go to IRQ1 service routine */

/* Instructions that check for other hardware interruptsigti be placed here */
OTHER_EXCEPTIONS:
* Instructions that check for other types of exceptionsudtitoe placed here */

.org 0x100
I* Interrupt-service routine for the desired hardwarerintpt */
EXT_IRQ1:

[* Instructions that handle the irgl interrupt request dtide placed here */

eret  /* Return from exception */

Figure 7. Code used to handle a hardware interrupt.

10 Cache Memory

As shown in Figure 4, a Nios Il system can include instructoil data caches, which are implemented in the
memory blocks in the FPGA chip. The caches can be specifieth @wlsystem is being designed by using the
SOPC Builder software. Inclusion of caches improves théopmance of a Nios Il system significantly, particu-
larly when most of the main memory is provided by an exterdaR8M chip, as is the case with Altera’s DE2
board. Both instruction and data caches are direct-mapped.

The instruction cache can be implemented in the fast andiatdrversions of the Nios Il processor systems.
Itis organized in 8 words per cache line, and its size is a-ssleictable design parameter.

The data cache can be implemented only with the Nios Il/f @seor. It has a configurable line size of 4, 16
or 32 bytes per cache line. Its overall size is also a usectsile design parameter.

10.1 Cache Management

Cache management is handled by software. For this purpesHitis Il instruction set includes the following
instructions:

e initd IMMED16(rA) (Initialize data-cache line)
Invalidates the line in the data cache that is associateld thvé address determined by adding the sign-
extended value IMMED16 and the contents of regisfer

e initi rA (Initialize instruction-cache line)
Invalidates the line in the instruction cache that is asgediwith the address contained in register

e flushd IMMED16(rA) (Flush data-cache line)
Computes the effective address by adding the sign-extevalad IMMED16 and the contents of register
rA. Then, it identifies the cache line associated with thisaive address, writes any dirty data in the cache
line back to memory, and invalidates the cache line.

e flushi rA (Flush instruction-cache line)
Invalidates the line in the instruction cache that is asgediwith the address contained in register

10.2 Cache Bypass Methods

A Nios Il processor uses its data cache in the standard maBugy it also allows the cache to be bypassed in
two ways. As mentioned in section 6.1, the Load and Storelasbns have a version intended for accessing I/O
devices, where the effective address specifies a locatian IfO device interface. These instructions dgevio,
Idbio, |duio, Idhio, Idhuio, stwio, stbio, andsthio. They bypass the data cache.

Another way of bypassing the data cache is by using bit 31 afdairess as a tag that indicates whether the
processor should transfer the data to/from the cache, adsyid. This feature is available only in the Nios II/f
processor.

Mixing cached and uncached accesses has to be done withQtherwise, the coherence of the cached data
may be compromised.

11 Tightly Coupled Memory

As explained in section 4, a Nios Il processor can access #imary blocks in the FPGA chip adightly coupled
memory This arrangement does not use the Avalon network. Insteadjghtly coupled memory is connected
directly to the processor.

Data in the tightly coupled memory is accessed using the abliwad and Store instructions, suchldw or
stw. The Nios Il control circuits determine if the address of amoey location is in the tightly coupled memory.
Accesses to the tightly coupled memory bypass the cachesh&address span of the tightly coupled memory,
the processor operates as if caches were not present.

